Seventh Framework Programme Theme 6 Environment

Collaborative Project (Large-scale Integrating Project)

Project no. **212085**

Project acronym: **MEECE**

Project title: **Marine Ecosystem Evolution in a Changing Environment**

Deliverable D2.5: Sub-model Carbon Phytoplankton including user guide

Statement: This document, of 04.11.2010, should be considered a replacement of the original document *D2.5 Sub-model for Carbon Enhancement in Marine Diatoms relative to pCO_2*, of 03.02.2010.

Due date of deliverable: August 2009
Actual submission date: September 2009, **revised November 2010**
Organisation name of lead contractor for this deliverable: UiB

Start date of project: 01.09.08 Duration: 48 months
Project Coordinator: Icarus Allen, Plymouth Marine Laboratory

Project co-funded by the European Commission within the Seventh Framework Programme, Theme 6 Environment		
Dissemination Level		
PU	Public	x
PP	Restricted to other programme participants (including the Commission)	
RE	Restricted to a group specified by the consortium (including the Commission)	
CO	Confidential, only for members of the consortium (including the Commission)	

D2.5 Sub-model Carbon Phytoplankton including user guide

R. Bellerby, G. Nondal, M. Butenschön, Y. Artioli, and J. Blackford

04.11.2010

Table of Contents

1. INTRODUCTION

Increasing inorganic carbon concentrations have been shown to promote primary production and carbon assimilation in marine phytoplankton (Hein and Sand-Jensen, 1997; Riebesell, 2004; Engel et al., 2008; Tortell et al., 2008; Egge et al., 2009), an effect rarely represented in ecosystem models. Here we present a module allowing for such carbon enhancement. As an example, the module is developed for the European Regional Seas Ecosystem Model (ERSEM; Blackford et al. 2004), a MEECE model, and we provide the necessary technical information aiding the inclusion of this module in other ecosystem models as well.

ERSEM is a mature plankton functional type model related to, but more complex than, the NPZD type models. It represents the key processes of temperate shelf ecosystems; plankton community complexity, the microbial loop, variable nutrient stoichiometry, variable carbon: chlorophyll ratios, and a comprehensive description of benthic biochemical and ecological processes (http://web.pml.ac.uk/meece/library/ersem.html). Droop kinetics is employed in evaluating nutrient limitation (Droop 1974), and separate limitation factors are calculated for each phytoplankton group (Blackford et al. 2004). We have modified a version of ERSEM coupled to the 1D General Ocean Turbulence Model (GOTM; http://www.gotm.net/), providing information on Temperature (T), Salinity (S), and mixing.

1.1 Representation of phytoplankton carbon enhancement in models

We extend the relationship describing the rate of change of a planktonic functional type in ERSEM (Blackford et al. 2004), and related models (Vichi et al. 2007), to include increased organic carbon production with increasing CO_2:

$$\frac{dP}{dt} = photosynthesis - respiration - lysis - excretion - grazing + \boldsymbol{CO_2\ enhancement} \qquad (1)$$

$$photosynthesis = r_{ass} \oplus t \oplus f_i \oplus P^C \oplus \mathbf{C^{EN}} \qquad (2)$$

where $\mathbf{C^{EN}}$**, the carbon enhancement factor,** relative to the year 2005, is represented as:

$$\mathbf{C^{EN} = (\mathit{p}CO_2^{yr} - \mathit{p}CO_2^{2005}) * \alpha^{EN} + 1}, \text{ where } \alpha^{EN} = 0.0005 \qquad (3)$$

As an example of potential use, the parameterization of C^{EN} follows from the Pelagic Ecosystem CO_2 Enrichment Study III (PeECE III) mesocosm experiments (Riebesell et al., 2007; Bellerby et al., 2008) where C^{EN} is related to the initial atmospheric CO_2 concentration at the onset of the bloom. However, the module is prepared to easily accommodate other forcing mechanisms from the meta-analysis of the MEECE dataset.

2. IMPLEMENTATION AND USER GUIDE

We use diatoms as a showcase example of the carbon-enhancement process in ERSEM, but it can be straightforwardly adapted to any plankton functional group, and model. Requiring relatively moderate modifications in ERSEM, the inclusion of carbon enhancement is implemented directly into the model code itself, and activated from an input file through an integer runtime switch. This procedure is, in the same way, applicable to other model systems. In the case of FORTRAN90, this is achieved by:

```
select case (switch)
   case(1)
     lines of code1
   case(2)
     lines of code2
end select
```

If including larger amounts of code, it might be beneficial to utilize conditional compilation, as represented in FORTRAN90 as:

```
#ifdef flag1
      lines of code
#endif
```

In this way the code will only be added to the executable if the pre-compilation flag, *flag1,* is defined, resulting in a smaller executable.

We will in the following *provide information regarding the specific processes requiring modification* in ERSEM, and show the specific modifications performed.

Affected processes – As shown in (1), the primary production of the respective functional group with regards to carbon uptake must be modified. The process of nitrogen and phosphorus uptake should remain unchanged (Riebesell et al. 2007, Bellerby et al. 2008).

Equation (1) can be written, in flux form (using the same notation as Vichi et al. 2007), as:

$$\left.\frac{\partial P_c}{\partial t}\right|_b = \left[\frac{\partial P_c}{\partial t}\right]^{pp} - \left[\frac{\partial P_c}{\partial t}\right]^{rsp}_{o3} - \left[\frac{\partial P_c}{\partial t}\right]^{out}_{R1} - \left[\frac{\partial P_c}{\partial t}\right]^{out}_{R6} \pm \sum_{Zy}\left[\frac{\partial P_c}{\partial t}\right]^{prd}_{Zy} + \left[\frac{\partial P_c}{\partial t}\right]^{\sin k} \quad (4)$$

where the left side of the equation distinguishes the fluxes that contribute to the biological rate of change. The last term can be positive or negative depending on the vertical gradient. We have to alter the gross primary production (PPD), as this mediates transfer of inorganic CO_2 to the plankton. This is described by:

$$\left[\frac{\partial P_c}{\partial t}\right]^{pp} = r_P P_c = \left(f_p^T f_p^I r_{0p}\right) P_c \qquad (5)$$

When describing Diatoms, functional group 1 in ERSEM, Equation (5), thus takes the form:

$$\left[\frac{\partial P1_c}{\partial t}\right]^{pp} = r_{P1} P1_c = \left(f_{p1}^T f_{p1}^I r_{0p1}\right) P1_c \qquad (6)$$

The constant parameter r_{0p} represents the maximum potential specific growth rate at the reference temperature (10°C). The process of carbon enhancement on gross production is therefore represented by:

$$\left[\frac{\partial P1_c}{\partial t}\right]^{pp} = r_{P1} P1_c = \left(f_{p1}^T f_{p1}^I r_{0p1}\right) P1_c \cdot C^{EN} \qquad (7)$$

In ERSEM, respiration is divided into basal – and activity respiration, as described by:

$$\left[\frac{\partial P_c}{\partial t}\right]_{O3}^{rsp} = \left(a_{P1} + b_P f_P^T\right) P_c \qquad (8)$$

The activity respiration $\left(a_{P1}\right)$ is a constant fraction of the assimilated carbon, which is in turn a function of the total rate of uptake minus the excretion loss. The basal respiration is dependent on the total growth rate, but modulated by the constant fraction of activity excretion, and the nutrient limitation. We thus need to enhance the activity respiration term in (8), as we have already altered the assimilation of carbon. This leads to the following expression:

$$\left[\frac{\partial P_c}{\partial t}\right]_{O3}^{rsp} = \left(a_{P1} \cdot C^{EN} + b_P f_P^T\right) P_c \qquad (9)$$

3. MODEL CODE

An example showing the necessary modifications for including a *plug-and-play carbon enhancement module* in ERSEM follows below. The modifications were performed in the original subroutine describing Diatom dynamics; vphyt1 (located in pprod.F90), more specifically, in the parts of the code handling (i) total respiration flux, and (ii) gross production as a flux from inorganic CO_2 to Diatoms. The *integer runtime switch,* called Cenhance, needs to be set in the input file gotmersem.nml, a file necessary for running the coupled model system. Cenhance = 1 represents a run with carbon enhancement, while Cenhance = 0 represents a standard run. This ensures the *plug-and-play feature*, as there is no need for changing any parameters in the code itself.

Statement of use: We strongly encourage the User to contact the originators of this document (R. Bellerby: Richard.bellerby@uni.no, G. Nondal: gislen@nersc.no), in order to receive supplementary information, prior to implementing the code.

Note: Changes to the original FORTRAN90 code are highlighted in blue and italics.

```
!-----------------------------------------------------------------------

  SUBROUTINE  vphyt1(I)
!
! !DESCRIPTION:
!  Diatom dynamics
!
  IMPLICIT NONE
!
!  INPUT VARIABLES:
 integer,intent(in) :: i
!  LOCAL VARIABLES:
 real*8 :: sdoP1,sumP1,sugP1,seoP1,seaP1,sraP1,sunP1,runP1,rugP1,sP1R6,runP1c
 real*8 :: runP1p,misP1p,rumP1p
 real*8 :: runP1n,misP1n,rumP1n,rumP1n3,rumP1n4
 real*8 :: etP1,SDP1,pe_R6P1
 real*8 :: rho,Chl_inc,Chl_loss
 real*8 :: phi,ChlCpp

#ifdef IRON
 real*8 :: runP1f,rumP1f,misP1f
#endif
!..Regulation factors.....................................................

!..Temperature response
      etP1 = q10P1X**((ETW(I)-10.0d0)/10.0d0) - q10P1X**((ETW(I)-32.0d0)/3.0d0)

!..Production.............................................................
```

```
!..calculate chl to C ratio.....................................

     ChlCpp = max(min(phimX,chl1(I)/(p1c(I)-chl1(I)+zeroX)),ChlCmin)

!..Gross photosynthetic activity :
#ifdef IRON
     sumP1 = sumP1X*etP1*iNP1s*iNP1f
#else
     sumP1 = sumP1X*etP1*iNP1s
#endif
     phi = phiHX + (ChlCpp/phimX)*(phimX-phiHX)

     parEIR = pEIR_eowX*EIR(I)
     IF (parEIR.gt.0.0) THEN
      sumP1 = sumP1 * (1.-exp(-alphaX*parEIR*ChlCpp/sumP1)) * EXP(-
betaX*parEIR*ChlCpp/sumP1)
       rho = phi * (sumP1/(alphaX*parEIR*ChlCpp))
     ELSE
      sumP1 = 0.0
      rho = 0.0
     END IF

!..Nutrient-stress lysis rate :
     sdoP1 = (1.d0/(iNIP1+0.1d0))*sdoP1X

!..Excretion rate, as regulated by nutrient-stress
     seoP1 = sumP1*(1.d0-iNIP1)*(1.d0-pu_eaP1X)

!..activity-dependent excretion :
     seaP1 = sumP1*pu_eaP1X
     sugP1 = sumP1-seoP1-seaP1

!..Apportioning of LOC- and DET- fraction of excretion/lysis fluxes:
     pe_R6P1 = MIN(qplP1cX/(qpP1c(I)+ZeroX),qnlP1cX/(qnP1c(I)+ZeroX))

     sP1R6 = pe_R6P1*sdoP1
     fP1R6c(I) = sP1R6*P1cP(I)

     fP1RDc(I) = (1.d0-pe_R6P1)*sdoP1*P1cP(I) + (seoP1 + seaP1)*P1c(I)

!..Respiration.....................................................

!..Rest respiration rate :
     srsP1 = etP1*srsP1X

!..Activity respiration rate :
     sraP1 = sugP1*pu_raP1X
```

```
SELECT CASE (Cenhance)

CASE(1)
! Changing carbon uptake as a function of pCO2 according to Bellerby et al. (2008) and
! Riebesell et al. (2007).

      pco2a5 = 379.48     !! setting atmospheric pCO2 to the level after which carbon
      !!enhancement occurs, here taken as  2005

      IF (PCO2A .GT. pco2a5) THEN
      cen=((PCO2A-pco2a5)*0.0005)+1. ! Carbon enhancement factor
      ELSE
      cen=1                                  ! no carbon enhancement
      ENDIF

!..Total respiration flux:
    fP1O3c(I) = srsP1*P1cP(I)+sraP1*P1c(I)*cen

!..Gross production as flux from inorganic CO2 to P1:
    fO3P1c(I) = sumP1*P1c(I)*cen
    rugP1 = fO3P1c(I)

CASE(0)
! standard run without C enhancement
!..Total respiration flux:
    fP1O3c(I) = srsP1*P1cP(I)+sraP1*P1c(I)

!..Gross production as flux from inorganic CO2 to P1:
    fO3P1c(I) = sumP1*P1c(I)
    rugP1 = fO3P1c(I)
CASE(2:)
    Write (6,*) 'ERROR: Cenhancement switch out of range!!'
END SELECT

!..Production and productivity
    sunP1 = sumP1-(seoP1+seaP1+sraP1)      ! net productivity
    runP1 = sunP1*P1c(I)-srsP1*P1cP(I)     ! net production

!..To save net production
    netP1(I) = runP1

!..Carbon Source equations

    rho = DMIN1(rho,0.1D0)
    ! Chl changes (note that Chl is a component of PXc and not involved
    ! in mass balance)
    Chl_inc = rho*(sumP1-sraP1)*P1c(I)
    Chl_loss = (sdoP1+srsP1)*Chl1P(I) + (seoP1+seaP1)*Chl1(I)
```

```
      SP1c(I) = SP1c(I)+fO3P1c(I)-fP1O3c(I)-fP1R6c(I)-fP1RDc(I)
      SR1c(I) = SR1c(I)+(fP1RDc(I) * R1R2X)
      SR2c(I) = SR2c(I)+(fP1RDc(I) * (1.0d0-R1R2X))
      SR6c(I) = SR6c(I)+fP1R6c(I)
      SChl1(I) = Schl1(I) + Chl_inc - Chl_loss

!..Phosphorus flux through P1.....

!..Lysis loss of phosphorus
      fP1R6p = sP1R6 * min(qplP1cX*P1cP(I),P1pP(I))
      fP1RDp = sdoP1 * P1pP(I) - fP1R6p

!..Net phosphorus uptake
      rumP1p = qurP1pX * N1pP(I) * P1c(I)
      misP1p = xqpP1X * qpRPIcX*P1cP(I) - P1pP(I)
      runP1p = runP1 * qpRPIcX*xqpP1X
      fN1P1p(I) = MIN(rumP1p, runP1p+misP1p)

!..Source equations
      SP1p(I) = SP1p(I)+fN1P1p(I)-fP1RDp-fP1R6p
      SN1p(I) = SN1p(I)-fN1P1p(I)
      SR6p(I) = SR6p(I)+fP1R6p
      SR1p(I) = SR1p(I)+fP1RDp

!..Nitrogen flux through P1.....

!..Nitrogen loss by lysis
      fP1R6n = sP1R6 * min(qnlP1cX*P1cP(I),P1nP(I))
      fP1RDn = sdoP1 * P1nP(I) - fP1R6n

!..Net nitrogen uptake

      rumP1n3 = quP1n3X * N3nP(I) * P1c(I)
      rumP1n4 = quP1n4X * N4nP(I) * P1c(I)
      rumP1n = rumP1n3 + rumP1n4

      misP1n = xqnP1X * qnRPIcX*P1cP(I) - P1nP(I)
      runP1n = runP1 * qnRPIcX*xqnP1X
      fNIP1n = MIN(rumP1n, runP1n + misP1n)

!..Partitioning over NH4 and NO3 uptake
      IF (fNIP1n .gt. 0.0d0) THEN
        fN3P1n(I) = fNIP1n * rumP1n3 / rumP1n
        fN4P1n(I) = fNIP1n * rumP1n4 / rumP1n
      ELSE
        fN3P1n(I) = 0.0d0
        fN4P1n(I) = fNIP1n
      ENDIF

!..Source equations
```

```
      SP1n(I) = SP1n(I)+fN4P1n(I)+fN3P1n(I)-fP1RDn-fP1R6n
      SN3n(I) = SN3n(I)-fN3P1n(I)
      SN4n(I) = SN4n(I)-fN4P1n(I)
      SR6n(I) = SR6n(I)+fP1R6n
      SR1n(I) = SR1n(I)+fP1RDn

!..Silicate flux through P1.....

!..Excretion loss of silicate
      fP1R6s = sdoP1 * P1s(I)

!..Loss of excess silicate (qsP1c > qsP1cX)
      fP1N5s(I) = MAX ( 0.0d0, P1sP(I)-qsP1cX * P1cP(I))

!..Net silicate uptake
      fN5P1s = MAX ( 0.0d0, qsP1cX*runP1) - fP1N5s(I)

!..Source equations
      SP1s(I) = SP1s(I) + fN5P1s - fP1R6s
      SN5s(I) = SN5s(I) - fN5P1s
      SR6s(I) = SR6s(I) + fP1R6s

#ifdef IRON
!..Iron flux through P1.....

!..Iron loss by lysis
!  Because its high affinity with particles all the iron lost from phytoplankton by lysis is supposed to
be
!  associated to organic particulate detritus. (luca)

      fP1R6f = sP1R6 * P1fP(I)

!..net iron uptake
      rumP1f = qurP1fX * N7fP(I) * P1c(I)
      misP1f = qfRP1cX*P1cP(I) - P1fP(I)
      runP1f = runP1 * qfRP1cX
      fN7P1f(I) = MIN(rumP1f, runP1f+misP1f)

!..Source equations
      SP1f(I) = SP1f(I)+fN7P1f(I)-fP1R6f
      SN7f(I) = SN7f(I)-fN7P1f(I)
      SR6f(I) = SR6f(I)+fP1R6f
#endif

!..sedimentation and resting stages.....

      SDP1 = resP1mX * MAX(0.d0, (esNIP1X - iNIP1) )
      SDP1c(I) = SDP1c(I) + SDP1
      SDP1p(I) = SDP1p(I) + SDP1
      SDP1n(I) = SDP1n(I) + SDP1
```

```
      SDP1s(I) = SDP1s(I) + SDP1
      SDChl1(I) = SDChl1(I) + SDP1
#ifdef IRON
      SDP1f(I) = SDP1f(I) + SDP1
#endif

!..add dia
!      IF ((P1c(I) .LT. 0.2).AND. (parEIR .GT. 0.5)) THEN
!       SP1c(I) = SP1c(I) + (0.2 - P1c(I))
!       SP1n(I) = SP1n(I) + (0.2 - P1c(I))*16.0/106.0
!       SP1p(I) = SP1p(I) + (0.2 - P1c(I))/106.0
!       SP1s(I) = SP1s(I) + (0.2 - P1c(I))*16.0/106.0
!       SChl1(I) = SChl1(I) + (0.2 - P1c(I))*ChlCpp
!      END IF

      RETURN

   END SUBROUTINE vphyt1
!-----------------------------------------------------------------------
```

3.1 Model results and discussion

After implementing the changes, as described above, the ERSEM-GOTM model pairing was run for the L4 site (50º15'N, 04º13'W) outside of Plymouth, UK, for three different experiments, each run for 3 years: (i) with an atmospheric CO_2 concentration of 350ppm, (ii) 700ppm and (iii) 1050ppm. Daily mean model output for the total water column of L4 was used. A Hovmöller plot of the C:N ratio ($R_{C:N}$) for the control run, together with the difference between $R_{C:N}$ at carbon enhancement and control are shown in figure 1.



Fig. 1 *Hovmöller plot showing Diatom C (mmol m^{-3}) vs. N (mmol m^{-3}) at the L4 site for experiment (i: top plot), the difference in $R_{C:N}$ between experiments (ii) and (i) is shown in the middle plot and the difference between experiments (iii) and (i) is shown in the bottom plot.*

It is evident from figure 1 that there has been a significant increase in $R_{C:N}$ from the control run to the carbon enhancement runs. It is further evident from figure 2 that the diatoms are assimilating more carbon at higher CO_2, as also seen in figure 3. Figure 3 also suggests that the Diatom bloom starts sooner in the high CO_2 runs, as can be explained by the higher availability of carbon.



Fig. 2 *Hovmöller plot showing Diatom C (mmol m^{-3}) at the L4 site for experiment (i; top plot), (ii; middle plot) and (iii; bottom plot).*



Fig. 3 *Plot showing the daily mean carbon concentrations in diatoms (mmol m^{-3}) for the three experiments: (i) 350ppm CO_2 (black), (ii) 700ppm CO_2 (red) and 1050ppm CO_2 (blue).*

Conclusions

We have shown that the *carbon enhancement module works for ERSEM*. The model specific feedbacks to enhanced carbon assimilation will be further investigated through WP3, together with further assessment of the proposed parameterisation.

4. References

Bellerby R. G. J. , Schulz K. G. , Riebesell U. , Neill C. , Nondal G. , Heegaard E. , Johannessen T. and Brown K. R., 2008. Marine ecosystem community carbon and nutrient uptake stoichiometry under varying ocean acidification during the PeECE III experiment Biogeosciences, 5, 1517-1527.

Blackford, J. C., Allen, J. I. and Gilbert, F. J. (2004) Ecosystem dynamics at size contrasting sites: a generic modelling study. J. Mar. Syst., 52, 191–215.

Droop, M.R., (1974). The nutrient status of algal cells in continuous culture. Journal of the Marine Biological Association of the UK 54, pp. 825–855.

Egge J. K., Thingstad T. F., Engel A., Bellerby R.G.J**,** and Riebesell U. (2009). Primary production during nutrient-induced blooms at elevated CO_2 concentrations. Biogeosciences, 6, 877-885.

Engel, A., Schulz, K.G., Riebesell, U., Bellerby R.G.J., Delille, B., Schartau, M. (2008). Effects of CO_2 on particle size distribution and phytoplankton abundance during a mesocosm bloom experiment (PeECE II). Biogeosciences, 5, 509–521.

Hein M. and Sand-Jensen K. (1997). CO_2 Increases Oceanic Primary Production. Nature, 388(6642), 526-527,.

Riebesell U., Schulz K. G., Bellerby R. G. J., Botros M., Fritsche P., Meyerhöfer M. Neill C., Nondal G., Oschlies A. and Wohlers J. (2007). Enhanced biological carbon consumption in a high CO_2 ocean. Nature, 450:545-549.

Vichi, M., Pinardi, N. and Masina, S. (2007). A generalized model of pelagic biogeochemistry for the global ocean ecosystem. Part I: Theory. JMS 64(2007) 89-109.